Mathematics 3530 Spring 2000
Lab 4

1 Purpose

The purpose of this lab is to use GAP to explore cosets.

All of the groups discussed in this lab will be subgroups of various symmetric groups. There is one important difference between the way GAP treats permutations and the way the author of our text does - permutations are multiplied in the opposite order in the two sources. For example, if f is the permutation given by $(1,2)(3,5)$ and g is the permutation given by $(1,2,3)$, the product of first f and then g in GAP is written as

```
(1,2)(3,5)*(1,2,3);
```

The product `(1,3,5)` is returned because f takes 1 to 2 and then g takes 2 to 3, so the composition $f * g$ takes 1 to 3, etc. Notice also that multiplication of non-disjoint cycles is indicated by *.

2 Signing on to GAP

- To boot Linux, place the Linux boot disk in the floppy drive, press Ctrl, Alt, and Del at the same time, click on Shutdown on the next window, and click on Shutdown and Restart on the next window. When the boot process has ended you will see something like

  ```
  Panther2178 login:
  ```

  In response to this, enter your username. (You do not need a password for your account the first time you sign on, but after signing on, you should change your password by issuing the command

  ```
  passwd
  ```

  and proceeding as directed.)

- To obtain a copy of the GAP lab for your own use, enter

  ```
  cp -r /home1/mat3530/lab4 .
  ```

  Be sure you type this command exactly as shown including the space between `lab4` and the period at the end of the command. This command will copy all of the relevant files to your account. *If you need to return to the lab at a later time to work on this lab, you do not need to get another copy (and you do not need to log on to the same computer).*

- To change to the `lab4` folder of your account enter

  ```
  cd lab4
  ```

- To start the X Windows environment enter

  ```
  startx
  ```

  Eventually you will get a screen with a K in the lower left corner. (Respond to any warning messages which appear by clicking on the `Oops` button.)

- To start an XGAP session find the icon on the lower menu which looks like a computer monitor with an attached screen and single click on it. Once a terminal window appears enter the command

  ```
  xgap &
  ```

  at the command line. When the window labelled `xgap` appears you are ready to go.

- To access the on-line help, start another terminal session as before. In the resulting terminal window enter

  ```
  netscape &
  ```

  at the command line. Once Netscape appears, go to the following URL

  ```
  http://www.ux1.eiu.edu/~cfdmb/gap/htm
  ```

  This connects you to on-line documentation. You can bounce between the xgap window and the Netscape window by moving the cursor to the appropriate window and left-clicking.

- To keep a copy of your GAP session enter

  ```
  LogTo("sessionx.gap");
  ```

  This causes your entire session to be recorded in the file `sessionx.gap` which you can edit later on. At the end of the session you can issue the command

  ```
  LogTo();
  ```

  which stops the logging to the file.

- To leave GAP by enter

  ```
  quit;
  ```

- To open a terminal session and edit the file you have created, enter

  ```
  emacs sessionx.gap &
  ```

  You can edit the file as desired, adding comments as you wish, and then print out the finished product and hand it in.

- When you have finished for the day, remove the Linux bootdisk from the disk drive. Move the cursor to the K in the lower left corner and press the left mouse button. Holding the button down, select Logout from the resulting menu and then release the button. Eventually you will exit the X Windows environment, although there may be some intervening windows. After exiting, press the Ctrl-Alt-Del key combination to reboot.

3 Cosets in Symmetric Groups

3.1 Defining Groups

First we need to define the groups. The symmetric group of degree 4 is generated by the permutations (1234) and (12). To define this group is GAP enter

```
S4 := Group([ (1,2,3,4),(1,2) ]);
```

To see the elements of this group and to verify that this group has 24 elements enter

```
Elements(S4);
```

and

```
Size(S4);
```

Now define K to be the subgroup of the symmetric group of degree 4 generated by (13)(24) and (14)(23) by

```
K := Subgroup(S4, [ (1,3)(2,4), (1,4)(2,3) ]);
```

Exercise 1.

a) Define H to be the subgroup of the symmetric group of degree 4 generated by (1234) and verify that $|K| = 4$.

b) Define T to be the subgroup of the symmetric group of degree 4 generated by (123) and verify that $|T| = 3$.

c) Define U to be the subgroup of the symmetric group of degree 4 generated by (1234) and (14)(23) and verify that $|U| = 8$.

d) Define A to be the subgroup of the symmetric group of degree 4 generated by (123) and (134) and verify that $|A| = 12$.

3.2 Cosets of K in S_4.

Now read in the file which contains the functions necessary for this lab by the command

```
Read("lab4functions.gap");
```

The group K, defined above, is normal in S_4. This can be verified with the following command

```
IsNormal(S4,K);
```

The output from this command should by `true`. Therefore, the set of right cosets of K in S_4 is a group. Since S_4 has 24 elements and K has 4, there are 6 elements in this group. To see what these elements are we need to select one "representative" from each coset. The command

```
RightCosetRepresentatives(S4,K);
```

Since this command selects, at random, one element from each coset your output will probably be different from what I obtained. My output which was

```
(1,2)(3,4)
(1,2)
(1,4)
(1,4,3)
(2,4,3)
(1,3)
```

According to this output,

$$S4/K = \{K(12)(34), K(12), K(14), K(143), K(243), K(13)\}$$

The elements in each of the six cosets can now be listed using the following six commands.

```
Elements(RightCoset(K,(1,2)(3,4)));
Elements(RightCoset(K,(1,2)));
Elements(RightCoset(K,(1,4)));
Elements(RightCoset(K,(1,4,3)));
Elements(RightCoset(K,(2,4,3)));
Elements(RightCoset(K,(1,3)));
```

Exercise 2. If you have not already done so, determine representatives for each of the cosets using `RightCosetRepresentatives` and using `Elements` and `RightCoset` determine the elements in each of the six cosets. (Your input will be very similar to the input above, but since your representatives may be different, your input will probably not be exactly the same.)

3.3 The Group S_4/K

By virtue of the last exercise, we know the elements of the group S_4/K. The operation of this group is coset multiplication. I have written the function `CosetProduct` to perform this operation, for example, the product $K(12)K(12)$ is computed by

```
CosetProduct(K,(1,2),(1,2));
```

From the output of this command you should see that this product is the coset K (alias $K(12)(34)$ to give just one of the other names for this coset).

Exercise 3.

a) Using the function `CosetProduct` find an element g such that $K(123)K(123) = Kg$.

b) Using the function `CosetProduct` verify that $K(123)Kg = K$.

c) Find the order of the element $K(123)$ in the group S_4/K.

d) Using the function `CosetProduct` verify that the group S_4/K is not abelian by showing

$$K(143)K(12) \neq K(12)K(143).$$

3.4 Cosets of H in S_4.

The group H is not normal in S_4 as the following command shows

```
IsNormal(S4,H);
```

However, the right cosets of H in S_4 still will partition the group into 6 cosets each containing 4 elements.

These could be found as above or, more simply, they can be printed out by the following command

```
PrintRightCosets(S4,H);
```

These cosets can still be multiplied together, however some of these products are cosets and some are not. For example, the following command shows that $H(24)H(234) = H(34)$

```
CosetProduct(H,(2,4),(2,3,4));
```

while the following command shows that the product $H(123)H(234)$ is not a coset!

```
CosetProduct(H,(1,2,3),(2,3,4));
```

Exercise 4.

a) Find another example of two cosets Hg_1 and Hg_2 such that the product Hg_1Hg_2 is a coset.

b) Find another example of two cosets Hg_3 and Hg_4 such that the product Hg_3Hg_4 is not a coset.

3.5 Cosets of T, U, and A in S_4

Exercise 5. For each of the subgroups T, U, and A,

a) determine the right cosets of the subgroup in S_4,

b) determine which of the subgroups are normal in S_4, and

c) for each subgroups that is not normal, find two cosets whose product is not a coset.

3.6 The group D_{12}

The group D_{12} is the group of symmetries of a regular hexagon. This group can be defined by the following command

```
D12 := Group([(1,2,3,4,5,6),(1,6)(2,5)(3,4)]);
```

The subgroup V of D_{12} that is generated by (14)(25)(34) is defined by

```
V := Subgroup(D12,[(1,4)(2,5)(3,6)]);
```

Exercise 6.

a) Verify that V is a normal subgroup of D_{12}.

b) Print out the six right cosets of V in D_{12}.

c) Determine the identity (as a coset) of D_{12}/V.

d) Determine the three elements (as cosets) of the group D_{12}/V which have order 2.

e) Determine the two elements (as cosets) of the group D_{12}/V which have order 3.

f) Select one element of D_{12}/V of order two, say Vk, and one element of D_{12}/V of order three, say Vt, and show the following relations are satisfied

- $(Vk)^2 = V$.
- $(Vt)^3 = V$.
- $(Vk)^{-1}(Vt)(Vk) = Vt^2$

(This will show D_{12}/V is isomorphic to S_3. Do you see why?)

4 What To Hand In

You are to hand in a completed, printed, edited copy of your work for this lab by Friday, April 14.

5 Appendix

Contents of `lab4functions.gap`

```
# Function to convert a list/group/set to a set
Convert := function(A)
   local Aset;
   if IsSet(A) then
      Aset := A;
   else
      Aset := AsSet(A);
   fi;
   return Aset;
end;

# Function to compute the product the sets A and B.
SetMultiply := function(A, B)
   local Aset, Bset, x, y, Pset;
   Aset := Convert(A);
   Bset := Convert(B);
   Pset := [];
   for x in Aset do
      for y in Bset do
         AddSet(Pset, x*y);
      od;
   od;
   return Pset;
end;

# Function to return a random element from each coset as a
#  representative
RightCosetRepresentatives := function(G,H)
   local k;
   for k in RightCosets(G,H) do
      Print( Random(k), "\n");
   od;
end;

# Function to multiply two right cosets together
# Computes the product (Hg)(Hk)
CosetProduct := function(H, g, k)
   return SetMultiply(RightCoset(H,g),RightCoset(H,k));
end;

# Function to print out right cosets
PrintRightCosets := function(G, H)
   local k;
   for k in RightCosets(G,H) do
      Print(Elements(k));
      Print("\n");
   od;
end;
```